// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _ownerOf(tokenId);
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner or approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(address from, address to, uint256 tokenId) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory data) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(address from, address to, uint256 tokenId, bytes memory data) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns the owner of the `tokenId`. Does NOT revert if token doesn't exist
     */
    function _ownerOf(uint256 tokenId) internal view virtual returns (address) {
        return _owners[tokenId];
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _ownerOf(tokenId) != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(address to, uint256 tokenId, bytes memory data) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId, 1);

        // Check that tokenId was not minted by `_beforeTokenTransfer` hook
        require(!_exists(tokenId), "ERC721: token already minted");

        unchecked {
            // Will not overflow unless all 2**256 token ids are minted to the same owner.
            // Given that tokens are minted one by one, it is impossible in practice that
            // this ever happens. Might change if we allow batch minting.
            // The ERC fails to describe this case.
            _balances[to] += 1;
        }

        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId, 1);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     * This is an internal function that does not check if the sender is authorized to operate on the token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId, 1);

        // Update ownership in case tokenId was transferred by `_beforeTokenTransfer` hook
        owner = ERC721.ownerOf(tokenId);

        // Clear approvals
        delete _tokenApprovals[tokenId];

        unchecked {
            // Cannot overflow, as that would require more tokens to be burned/transferred
            // out than the owner initially received through minting and transferring in.
            _balances[owner] -= 1;
        }
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId, 1);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(address from, address to, uint256 tokenId) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId, 1);

        // Check that tokenId was not transferred by `_beforeTokenTransfer` hook
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");

        // Clear approvals from the previous owner
        delete _tokenApprovals[tokenId];

        unchecked {
            // `_balances[from]` cannot overflow for the same reason as described in `_burn`:
            // `from`'s balance is the number of token held, which is at least one before the current
            // transfer.
            // `_balances[to]` could overflow in the conditions described in `_mint`. That would require
            // all 2**256 token ids to be minted, which in practice is impossible.
            _balances[from] -= 1;
            _balances[to] += 1;
        }
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId, 1);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(address owner, address operator, bool approved) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens will be transferred to `to`.
     * - When `from` is zero, the tokens will be minted for `to`.
     * - When `to` is zero, ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal virtual {}

    /**
     * @dev Hook that is called after any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens were transferred to `to`.
     * - When `from` is zero, the tokens were minted for `to`.
     * - When `to` is zero, ``from``'s tokens were burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal virtual {}

    /**
     * @dev Unsafe write access to the balances, used by extensions that "mint" tokens using an {ownerOf} override.
     *
     * WARNING: Anyone calling this MUST ensure that the balances remain consistent with the ownership. The invariant
     * being that for any address `a` the value returned by `balanceOf(a)` must be equal to the number of tokens such
     * that `ownerOf(tokenId)` is `a`.
     */
    // solhint-disable-next-line func-name-mixedcase
    function __unsafe_increaseBalance(address account, uint256 amount) internal {
        _balances[account] += amount;
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                // The surrounding unchecked block does not change this fact.
                // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1, "Math: mulDiv overflow");

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10 ** 64) {
                value /= 10 ** 64;
                result += 64;
            }
            if (value >= 10 ** 32) {
                value /= 10 ** 32;
                result += 32;
            }
            if (value >= 10 ** 16) {
                value /= 10 ** 16;
                result += 16;
            }
            if (value >= 10 ** 8) {
                value /= 10 ** 8;
                result += 8;
            }
            if (value >= 10 ** 4) {
                value /= 10 ** 4;
                result += 4;
            }
            if (value >= 10 ** 2) {
                value /= 10 ** 2;
                result += 2;
            }
            if (value >= 10 ** 1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/math/SignedMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SignedMath.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard signed math utilities missing in the Solidity language.
 */
library SignedMath {
    /**
     * @dev Returns the largest of two signed numbers.
     */
    function max(int256 a, int256 b) internal pure returns (int256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two signed numbers.
     */
    function min(int256 a, int256 b) internal pure returns (int256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two signed numbers without overflow.
     * The result is rounded towards zero.
     */
    function average(int256 a, int256 b) internal pure returns (int256) {
        // Formula from the book "Hacker's Delight"
        int256 x = (a & b) + ((a ^ b) >> 1);
        return x + (int256(uint256(x) >> 255) & (a ^ b));
    }

    /**
     * @dev Returns the absolute unsigned value of a signed value.
     */
    function abs(int256 n) internal pure returns (uint256) {
        unchecked {
            // must be unchecked in order to support `n = type(int256).min`
            return uint256(n >= 0 ? n : -n);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";
import "./math/SignedMath.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `int256` to its ASCII `string` decimal representation.
     */
    function toString(int256 value) internal pure returns (string memory) {
        return string(abi.encodePacked(value < 0 ? "-" : "", toString(SignedMath.abs(value))));
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }

    /**
     * @dev Returns true if the two strings are equal.
     */
    function equal(string memory a, string memory b) internal pure returns (bool) {
        return keccak256(bytes(a)) == keccak256(bytes(b));
    }
}

// File: contracts/DecentralName.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.9;

/*
    Errors:
    E1: No authority
    E2: Invalid namespace
    E3: Invalid name 
    E4: Name expired
    E5: Array count mismatch
    E6: Fee not enough
    E7: Namespace already exists
    E8: Cannot modify price
    E9: Invalid price
    E10: Too many custom price names
    E11: Already open
    E12: Already disabled
    E13: Not open yet
    E14: Cannot receive name
    E15: Not support ping
    E16: Content too long
    E17: Key mismatch
    E18: Invalid key
    E19: Too many keys
    E20: Transfer not allowed
    E21: Not need renewal
    E22: Not expired yet
    E23: Invalid lifetime
    E24: Invalid times
    E25: Transfer not allow: to whom already own non-expired name
    E26: Invalid renewal price param
*/

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC721/ERC721.sol";

contract DecentralName is ReentrancyGuard, ERC721 {
    uint256 constant public MAX_PROFILE_KEY_COUNT = 200;
    uint256 constant public MAX_CUSTOM_PRICE_NAMES_COUNT = 200;
    uint256 constant public MAX_RENEWAL_TIMES_PER_TIMES = 20;
    uint256[] public NAMESPACE_PRICE_LIST = [250 ether, 50 ether, 40 ether, 20 ether, 10 ether, 5 ether, 5 ether, 2 ether];

    event BuyNamespace(bytes32 indexed namespace);
    event TransferNamespaceAdmin(bytes32 indexed namespace, address admin);
    event SetNamespaceBeneficiary(bytes32 indexed namespace, address beneficiary);
    event SetNamespaceDesc(bytes32 indexed namespace);
    event SetPrice(bytes32 indexed namespace);
    event SetCustomPrice(bytes32 indexed namespace);
    event OpenNamespace(bytes32 indexed namespace);
    event DisableModifyPrice(bytes32 indexed namespace);
    event AssignName(bytes32 indexed namespace, bytes32 indexed name, address user);
    event Renewal(bytes32 indexed namespace, bytes32 indexed name);
    event Ping(bytes32 indexed namespace, bytes32 indexed name);
    event SetProfile(bytes32 indexed namespace, bytes32 indexed name);

    struct NamespaceInfo {
        address admin;                  // Who can modify the namespace's attributes
        address payable beneficiary;    // Where the registration/renewal fee goes to
        bytes desc;                     // Length <= 256, shown as innerHTML. e.g., "visit <a href="https://nostr.com" target="_blank">nostr.com</a> for more info".
        uint256 startStamp;             // When was the namespace created
        uint256 openStamp;              // When was the namespace open for registration(i.e., anyone can register through current contract)
        uint256 renewalLifetime;        // Renewal period, 0 means not need renewal.
        uint256 pingLifetime;           // Only makes sense when renewalLifetime == 0. If pingLifetime > 0, user must ping before pingExpiredStamp(only cost gas fee), otherwise the name is expired.
        uint256[25] registerPriceList;  // 1d, 1L, _, 2d, 2L, 2Misc, 3d, 3L, 3Misc, 4d, 4L, 4Misc, 5d, 5L, 5Misc, 6d, 6L, 6Misc, 7d, 7L, 7Misc, 8d, 8L, 8Misc, remain.
        uint256 renewalPriceParam;      // renewalPriceParam = a * 1000000 + b * 1000 + c, Renewal-fee = (a + b * Registration-fee) / c
        bool bAllowModifyPrice;         // Can namespace admin modify price?
        bytes32[] customPriceNames;     // Custom registration prices for at most MAX_CUSTOM_PRICE_NAMES_COUNT names.
    }

    struct NameInfo {
        bytes32 namespace;
        bytes32 name;
        uint256 renewalExpiredStamp;
        uint256 pingExpiredStamp;
        bytes32[] profileKeyList; // display name, avatar, bio, url, nostr, email, btc, lightning, ..
    }

    bytes32[] public namespaceList;
    mapping(bytes32 => NamespaceInfo) public namespaceMap;                      // namespace => NamespaceInfo
    mapping(bytes32 => mapping(bytes32 => uint256)) public customPriceNamesMap; // namespace => (name => custom-price)

    uint256 public curTokenId = 0;
    mapping(address => uint256) public addr2TokenIdMap;                         // address => tokenId
    mapping(uint256 => NameInfo) public id2NameInfoMap;                         // tokenId => NameInfo
    mapping(uint256 => mapping(bytes32 => bytes)) public id2ProfileValueMap;    // tokenId => (profile-key => profile-value)
    mapping(bytes32 => mapping(bytes32 => uint256)) public name2IdMap;          // (namespace => (name => tokenId))

    constructor() ERC721("DecentralName", "DNAME") {
        _initGenesis();
    }

    //////// Frequent
    function resolveAddress(address user) public view returns (bool bResult, uint256 tokenId, bytes32 namespace, bytes32 name) {
        tokenId = addr2TokenIdMap[user];
        if (tokenId > 0) {
            (bResult, namespace, name) = resolveTokenId(tokenId);
        }
    }

    function resolveAddressFull(address user) public view returns (bool bResult, uint256 tokenId, NameInfo memory nameInfo, bytes[] memory profileValueList) {
        (bResult, tokenId, nameInfo.namespace, nameInfo.name) = resolveAddress(user);
        if (bResult) {
            (, , profileValueList) = getProfileKeysAndValues(nameInfo.namespace, nameInfo.name);
            nameInfo = id2NameInfoMap[tokenId];
        }
    }

    function resolveName(bytes32 namespace, bytes32 name) public view returns (bool bResult, address owner) {
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId > 0) {
            owner = ownerOf(tokenId);
            NameInfo storage nInfo = id2NameInfoMap[tokenId];
            if (!_isNameExpired(namespaceMap[nInfo.namespace], nInfo)) {
                bResult = true;
            }
        }
    }

    function resolveNameFull(bytes32 namespace, bytes32 name) public view returns (bool bResult, uint256 tokenId, address owner, NameInfo memory nameInfo, bytes[] memory profileValueList) {
        (bResult, ,profileValueList) = getProfileKeysAndValues(namespace, name);
        if (bResult) {
            tokenId = name2IdMap[namespace][name];
            owner = ownerOf(tokenId);
            nameInfo = id2NameInfoMap[tokenId];
        }
    }

    // Query many names to register
    function queryManyNames(bytes32[] calldata inNamespaceList, bytes32[] calldata inNameList) public view returns (bool[] memory resultList, address[] memory ownerList, uint256[] memory priceList) {
        uint256 nameListLen = inNameList.length;
        require(nameListLen == inNamespaceList.length, 'E5');
        resultList = new bool[](nameListLen);
        ownerList = new address[](nameListLen);
        priceList = new uint256[](nameListLen);
        for (uint256 i; i < nameListLen; ++i) {
            uint256 tokenId = name2IdMap[inNamespaceList[i]][inNameList[i]];
            NamespaceInfo storage nsInfo = namespaceMap[inNamespaceList[i]];
            if (tokenId > 0 && !_isNameExpired(nsInfo, id2NameInfoMap[tokenId])) {
                resultList[i] = false;
                ownerList[i] = ownerOf(tokenId);
            } else if (nsInfo.openStamp > 0) {
                resultList[i] = true;
                (, priceList[i]) = getRegistrationPrice(inNamespaceList[i], inNameList[i]);
            }
        }
    }

    function resolveTokenId(uint256 tokenId) public view returns (bool bResult, bytes32 namespace, bytes32 name) {
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        NamespaceInfo storage nsInfo = namespaceMap[nInfo.namespace];
        if (nsInfo.startStamp > 0 && !_isNameExpired(nsInfo, nInfo)) {
            bResult = true;
            namespace = nInfo.namespace;
            name = nInfo.name;
        }
    }

    function getNamespaceInfos(bytes32[] memory inNamespaceList) public view returns (NamespaceInfo[] memory results) {
        uint256 listLen = inNamespaceList.length;
        results = new NamespaceInfo[](listLen);
        for (uint256 i; i < listLen; ++i) {
            results[i] = namespaceMap[inNamespaceList[i]];
        }
    }

    function getFullNamespaceInfos() public view returns (bytes32[] memory outNamespaceList, NamespaceInfo[] memory results) {
        outNamespaceList = namespaceList;
        results = getNamespaceInfos(namespaceList);
    }

    //////// Namespace
    function getNamespacePrice(bytes32 namespace) public view returns (bool bResult, uint256 price) {
        (bool bOk, uint256 alphaNum, uint256 digitNum, uint256 underscoreNum) = parseName(namespace);
        if (bOk && (namespace[0] >= 0x61 && namespace[0] <= 0x7a)) {  // For namespace, first character must be lower-case-alpha
            bResult = true;
            uint256 totalLen = alphaNum + digitNum + underscoreNum;
            if (totalLen <= NAMESPACE_PRICE_LIST.length) {
                price = NAMESPACE_PRICE_LIST[totalLen - 1];
            } else {
                price = NAMESPACE_PRICE_LIST[NAMESPACE_PRICE_LIST.length - 1];
            }
        }
    }

    function buyNamespace(bytes32 namespace, uint256 renewalLifetime, uint256 pingLifetime) payable external nonReentrant {
        (bool bOk, uint256 price) = getNamespacePrice(namespace);
        require(bOk, 'E2');
        require(renewalLifetime == 0 || (renewalLifetime >= 1 days && renewalLifetime <= 36500 days), 'E23');
        if (renewalLifetime > 0) {
            require(pingLifetime == 0, 'E23');
        } else {
            require(pingLifetime == 0 || (pingLifetime >= 1 days && pingLifetime <= 36500 days), 'E23');
        }
        require(msg.value == price, 'E6');
        payable(0).transfer(price); // burn

        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.startStamp == 0, 'E7');
        namespaceList.push(namespace);

        nsInfo.admin = msg.sender;
        nsInfo.beneficiary = payable(msg.sender);
        // nsInfo.desc
        nsInfo.startStamp = block.timestamp;
        // nsInfo.openStamp
        nsInfo.renewalLifetime = renewalLifetime;
        nsInfo.pingLifetime = pingLifetime;
        // nsInfo.registerPriceList
        // nsInfo.renewalPriceParam
        nsInfo.bAllowModifyPrice = true;
        // nsInfo.customPriceNames

        emit BuyNamespace(namespace);
    }

    function transferNamespaceAdmin(bytes32 namespace, address admin) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        nsInfo.admin = admin;
        emit TransferNamespaceAdmin(namespace, admin);
    }

    function setNamespaceBeneficiary(bytes32 namespace, address payable beneficiary) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender || nsInfo.beneficiary == msg.sender, 'E1');
        nsInfo.beneficiary = beneficiary;
        emit SetNamespaceBeneficiary(namespace, beneficiary);
    }

    function setNamespaceDesc(bytes32 namespace, bytes calldata desc) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        require(desc.length <= 256, 'E16');
        nsInfo.desc = desc;
        emit SetNamespaceDesc(namespace);
    }

    function setPrice(bytes32 namespace, uint256[25] calldata registerPriceList, uint256 renewalPriceParam) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        require(nsInfo.bAllowModifyPrice, 'E8');
        require(nsInfo.renewalLifetime == 0 || renewalPriceParam % 1000 != 0, 'E26');
        nsInfo.registerPriceList = registerPriceList;
        nsInfo.renewalPriceParam = renewalPriceParam;
        emit SetPrice(namespace);
    }

    function setCustomPrice(bytes32 namespace, bytes32[] calldata nameList, uint256[] calldata priceList) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        require(nsInfo.bAllowModifyPrice, 'E8');
        uint256 nameLen = nameList.length;
        require(nameLen == priceList.length, 'E5');
        bytes32[] storage customNames = nsInfo.customPriceNames;
        mapping(bytes32 => uint256) storage priceMap = customPriceNamesMap[namespace];
        for (uint256 i; i < nameLen; ) {
            bytes32 name = nameList[i];
            if (priceMap[name] == 0) {
                require(isNameValid(name), 'E3');
                customNames.push(name);
            }
            require(priceList[i] > 0, 'E9'); // i.e., not allow cancel set custom name price
            priceMap[name] = priceList[i];
            unchecked {
                ++i;
            }
        }
        require(customNames.length <= MAX_CUSTOM_PRICE_NAMES_COUNT, 'E10');
        emit SetCustomPrice(namespace);
    }

    function getCustomPriceInfo(bytes32 namespace) public view returns (bytes32[] memory nameList, uint256[] memory priceList) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        nameList = nsInfo.customPriceNames;
        uint256 nameListLen = nameList.length;
        priceList = new uint256[](nameListLen);
        mapping(bytes32 => uint256) storage priceMap = customPriceNamesMap[namespace];
        for (uint256 i; i < nameListLen; ++i) {
            priceList[i] = priceMap[nameList[i]];
        }
    }

    // Open registration, so that anyone can register through current contract; Otherwise, only admin can register through registerMany(with bByAdmin=true)
    function openRegistration(bytes32 namespace) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        require(nsInfo.openStamp == 0, 'E11');
        require(nsInfo.renewalLifetime == 0 || nsInfo.renewalPriceParam % 1000 != 0, 'E26');
        nsInfo.openStamp = block.timestamp;
        emit OpenNamespace(namespace);
    }

    function disableModifyPrice(bytes32 namespace) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.admin == msg.sender, 'E1');
        require(nsInfo.bAllowModifyPrice, 'E12');
        require(nsInfo.renewalLifetime == 0 || nsInfo.renewalPriceParam % 1000 != 0, 'E26');
        nsInfo.bAllowModifyPrice = false;
        emit DisableModifyPrice(namespace);
    }

    //////// Name
    function register(bytes32 namespace, bytes32 name) external payable nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.openStamp > 0, 'E13');
        (bool bCanReceiveName,) = _canReceiveName(msg.sender);
        require(bCanReceiveName, 'E14');
        (bool bOk, uint256 price) = getRegistrationPrice(namespace, name);
        require(bOk, 'E3');
        if (price > 0) {
            require(msg.value >= price, 'E6');
            nsInfo.beneficiary.transfer(msg.value);
        }
        _assignNameToUser(namespace, name, msg.sender);
    }

    // Force register name. If current address already owns non-expired name, burn it firstly
    function forceRegister(bytes32 namespace, bytes32 name) external payable nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.openStamp > 0, 'E13');
        (bool bCanReceiveName, uint256 tokenId) = _canReceiveName(msg.sender);
        if (!bCanReceiveName) {
            _removeItem(tokenId);
        }
        (bool bOk, uint256 price) = getRegistrationPrice(namespace, name);
        require(bOk, 'E3');
        if (price > 0) {
            require(msg.value >= price, 'E6');
            nsInfo.beneficiary.transfer(msg.value);
        }
        _assignNameToUser(namespace, name, msg.sender);
    }

    function registerMany(bool bByAdmin, bytes32 namespace, bytes32[] calldata nameList, address[] calldata userList) external payable nonReentrant returns (address[] memory failList, uint256 failCount) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        if (bByAdmin) {
            require(nsInfo.admin == msg.sender, 'E1');
        } else {
            require(nsInfo.openStamp > 0, 'E13');
        }
        
        uint256 nameLen = nameList.length;
        require(nameLen == userList.length, 'E5');
        failList = new address[](nameLen);
        uint256 totalFee = 0;
        for (uint256 i; i < nameLen; ) {
            require(userList[i] != address(0), 'E14');
            (bool bCanReceiveName,) = _canReceiveName(userList[i]);
            if (bCanReceiveName) {
                if (bByAdmin) {
                    require(isNameValid(nameList[i]), 'E3');
                } else {
                    (bool bResult, uint256 price) = getRegistrationPrice(namespace, nameList[i]);
                    require(bResult, 'E3');
                    totalFee += price;
                }
                _assignNameToUser(namespace, nameList[i], userList[i]);
            } else {
                failList[failCount++] = userList[i];
            }
            unchecked {
                ++i;
            }
        }

        if (totalFee > 0) {
            require(totalFee <= msg.value, 'E6');
            nsInfo.beneficiary.transfer(totalFee);
        }
        if (totalFee < msg.value) {
            payable(msg.sender).transfer(msg.value - totalFee);
        }
    }

    function renewal(bytes32 namespace, bytes32 name, uint256 times) external payable nonReentrant returns (bool bResult) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(times >= 1 && times <= MAX_RENEWAL_TIMES_PER_TIMES, 'E24');
        require(nsInfo.renewalLifetime > 0, 'E21');
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId > 0) {
            NameInfo storage nInfo = id2NameInfoMap[tokenId];
            if (!_isNameExpired(nsInfo, nInfo)) {
                (, uint256 renewalPrice) = getRenewalPrice(namespace, name);
                if (renewalPrice > 0) {
                    require(msg.value >= renewalPrice * times, 'E6');
                    nsInfo.beneficiary.transfer(msg.value);
                }
                nInfo.renewalExpiredStamp = nInfo.renewalExpiredStamp + times * nsInfo.renewalLifetime;
                emit Renewal(namespace, name);
                bResult = true;
            }
        }
    }

    function renewalMany(bool bByAdmin, bytes32 namespace, bytes32[] calldata nameList, uint256[] calldata timesList) external payable nonReentrant returns (bytes32[] memory failList, uint256 failCount) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        if (bByAdmin) {
            require(nsInfo.admin == msg.sender, 'E1');
        }
        require(nsInfo.renewalLifetime > 0, 'E21');
        uint256 nameLen = nameList.length;
        require(nameLen == timesList.length, 'E5');
        failList = new bytes32[](nameLen);
        uint256 totalFee = 0;
        for (uint256 i; i < nameLen; ) {
            bytes32 name = nameList[i];
            uint256 tokenId = name2IdMap[namespace][name];
            if (tokenId > 0) {
                NameInfo storage nInfo = id2NameInfoMap[tokenId];
                if (!_isNameExpired(nsInfo, nInfo)) {
                    if (!bByAdmin) {
                        (, uint256 renewalPrice) = getRenewalPrice(namespace, name);
                        totalFee += renewalPrice * timesList[i];
                    }
                    require(timesList[i] >= 1 && timesList[i] <= MAX_RENEWAL_TIMES_PER_TIMES, 'E24');
                    nInfo.renewalExpiredStamp = nInfo.renewalExpiredStamp + timesList[i] * nsInfo.renewalLifetime;
                    emit Renewal(namespace, name);
                } else {
                    failList[failCount++] = name;
                }
            } else {
                failList[failCount++] = name;
            }
            unchecked {
                ++i;
            }
        }

        if (totalFee > 0) {
            require(totalFee <= msg.value, 'E6');
            nsInfo.beneficiary.transfer(totalFee);
            if (totalFee < msg.value) {
                payable(msg.sender).transfer(msg.value - totalFee);
            }
        }
        return (failList, failCount);
    }

    function ping(bytes32 namespace, bytes32 name) external nonReentrant {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        require(nsInfo.renewalLifetime == 0 && nsInfo.pingLifetime > 0, 'E15');
        uint256 tokenId = name2IdMap[namespace][name];
        require(ownerOf(tokenId) == msg.sender, 'E1');
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        require(block.timestamp <= nInfo.pingExpiredStamp, 'E4');
        nInfo.pingExpiredStamp = block.timestamp + nsInfo.pingLifetime;
        emit Ping(namespace, name);
    }

    function getRegistrationPrice(bytes32 namespace, bytes32 name) public view returns (bool, uint256) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        if (nsInfo.startStamp == 0) {
            return (false, 0);
        }

        uint256 customPrice = customPriceNamesMap[namespace][name];
        if (customPrice > 0) {
            return (true, customPrice);
        }

        (bool bOk, uint256 alphaNum, uint256 digitNum, uint256 underscoreNum) = parseName(name);
        if (!bOk) {
            return (false, 0);
        }
        uint256 totalLen = alphaNum + digitNum + underscoreNum;
        if (totalLen <= 8) {
            if (underscoreNum == 0 && (alphaNum == 0 || digitNum == 0)) { // pure-alpha or pure-digit or empty
                if (digitNum > 0) {
                    return (true, nsInfo.registerPriceList[(totalLen - 1) * 3]);
                } else if (alphaNum > 0) {
                    return (true, nsInfo.registerPriceList[totalLen * 3 - 2]);
                } else {
                    return (false, 0);
                }
            } else { // Misc
                return (true, nsInfo.registerPriceList[totalLen * 3 - 1]);
            }
        }
        return (true, nsInfo.registerPriceList[nsInfo.registerPriceList.length - 1]);
    }

    function getRenewalPrice(bytes32 namespace, bytes32 name) public view returns (bool, uint256) {
        (bool bOk, uint256 regPrice) = getRegistrationPrice(namespace, name);
        if (bOk) {
            NamespaceInfo storage nsInfo = namespaceMap[namespace];
            if (nsInfo.renewalLifetime == 0) {
                return (true, 0);   // not need renewal
            }
            uint256 param = nsInfo.renewalPriceParam;
            return (true, ((param/1000000) + ((param/1000)%1000) * regPrice) / (param % 1000));
        }
        return (false, 0);
    }

    function parseName(bytes32 name) public pure returns (bool, uint256, uint256, uint256) {
        bool bReachEnd = false;
        uint256 alphaNum = 0;
        uint256 digitNum = 0;
        uint256 underscoreNum = 0;
        for (uint256 i; i < 32; ) {
            bytes1 ch = name[i];
            if (ch > 0) {
                if (bReachEnd) {
                    return (false, 0, 0, 0);
                }
                if (ch >= 0x61 && ch <= 0x7a) {
                    ++alphaNum;
                } else if (ch >= 0x30 && ch <= 0x39) {
                    ++digitNum;
                } else if (ch == 0x5f) {
                    ++underscoreNum;
                } else {
                    return (false, 0, 0, 0);
                }
            } else {
                if (!bReachEnd) {
                    bReachEnd = true;
                }
            }
            unchecked {
                ++i;
            }
        }

        if (alphaNum == 0 && digitNum == 0 && underscoreNum == 0) {
            return (false, 0, 0, 0);
        }
        return (true, alphaNum, digitNum, underscoreNum);
    }

    function canReceiveName(address user) public view returns (bool bCanReceiveName) {
        (bCanReceiveName, ) = _canReceiveName(user);
    }

    function canRegisterName(bytes32 namespace, bytes32 name) public view returns (bool) {
        return canRegisterName(namespace, name, true);
    }

    function canRegisterName(bytes32 namespace, bytes32 name, bool bCheckOpen) public view returns (bool) {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        if (nsInfo.startStamp == 0) {
            return false;
        }
        if (bCheckOpen && nsInfo.openStamp == 0) {
            return false;
        }
        if (!isNameValid(name)) {
            return false;
        }
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId == 0) {
            return true;
        }
        return _isNameExpired(nsInfo, id2NameInfoMap[tokenId]);
    }

    function isNameValid(bytes32 name) public pure returns (bool) {
        uint256 flag = 0;
        for (uint256 i = 0; i < 32; ++i) {
            bytes1 ch = name[i];
            if (ch > 0) {
                if (flag == 2) {
                    return false;
                } else {
                    flag = 1;
                }
                if ((ch < 0x61 || ch > 0x7a) && (ch < 0x30 || ch > 0x39) && ch != 0x5f) {  // not-lower-case-alpha && not-digit && not-underscore
                    return false;
                }
            } else {
                if (flag != 2) {
                    if (flag == 0) {
                        return false;
                    }
                    flag = 2;
                }
            }
        }
        return flag != 0;
    }

    // display name:    0x646973706c6179206e616d650000000000000000000000000000000000000000
    // avatar:          0x6176617461720000000000000000000000000000000000000000000000000000
    // bio:             0x62696f0000000000000000000000000000000000000000000000000000000000
    // url:             0x75726c0000000000000000000000000000000000000000000000000000000000
    // nostr:           0x6e6f737472000000000000000000000000000000000000000000000000000000
    // email:           0x656d61696c000000000000000000000000000000000000000000000000000000
    // btc:             0x6274630000000000000000000000000000000000000000000000000000000000
    // lightning:       0x6c696768746e696e670000000000000000000000000000000000000000000000
    function getProfileValue(bytes32 namespace, bytes32 name, bytes32 key) public view returns (bool bResult, bytes memory value) {
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId > 0) {
            if (!_isNameExpired(namespaceMap[namespace], id2NameInfoMap[tokenId])) {
                bResult = true;
                value = id2ProfileValueMap[tokenId][key];
            }
        }
    }

    function getProfileKeysAndValues(bytes32 namespace, bytes32 name) public view returns (bool bResult, bytes32[] memory keyList, bytes[] memory valueList) {
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId > 0) {
            NameInfo storage nInfo = id2NameInfoMap[tokenId];
            if (!_isNameExpired(namespaceMap[namespace], nInfo)) {
                bResult = true;
                keyList = nInfo.profileKeyList;
                uint256 keyListLen = keyList.length;
                valueList = new bytes[](keyListLen);
                mapping(bytes32 => bytes) storage profileValueMap = id2ProfileValueMap[tokenId];
                for (uint256 i; i < keyListLen; ++i) {
                    valueList[i] = profileValueMap[keyList[i]];
                }
            }
        }
    }

    // Set profile's custom key-value pairs.
    // If indexList[pos] > 0, set profileValueMap[keyList[indexList[pos]-1]] = valueList[pos];
    // Otherwise, append keyList[pos] to profileKeyList, set profileValueMap[keyList[pos]] = valueList[pos];
    function setProfileKeyValuePairs(bytes32 namespace, bytes32 name, uint256[] calldata indexList, bytes32[] calldata keyList, bytes[] calldata valueList) external nonReentrant {
        uint256 tokenId = name2IdMap[namespace][name];
        require(ownerOf(tokenId) == msg.sender, 'E1');
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        require(!_isNameExpired(namespaceMap[namespace], nInfo), 'E4');

        uint256 indexLen = indexList.length;
        require(indexLen == keyList.length, 'E5');
        require(indexLen == valueList.length, 'E5');
        mapping(bytes32 => bytes) storage profileValueMap = id2ProfileValueMap[tokenId];
        bytes32[] storage profileKeyList = nInfo.profileKeyList;
        for (uint256 pos; pos < indexLen; ) {
            require(valueList[pos].length <= 128, 'E16');
            if (indexList[pos] > 0) { // real index is curIndex - 1
                bytes32 curKey = profileKeyList[indexList[pos] - 1];
                require(curKey == keyList[pos], 'E17');
                profileValueMap[curKey] = valueList[pos];
            } else {
                bytes32 curKey = keyList[pos];
                profileKeyList.push(curKey);
                require(curKey != 0, 'E18');
                profileValueMap[curKey] = valueList[pos];
            }
            unchecked {
                ++pos;
            }
        }
        require(profileKeyList.length <= MAX_PROFILE_KEY_COUNT, 'E19');
        emit SetProfile(namespace, name);
    }

    function removeAllProfileKeys(bytes32 namespace, bytes32 name) external nonReentrant {
        uint256 tokenId = name2IdMap[namespace][name];
        require(ownerOf(tokenId) == msg.sender, 'E1');
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        require(!_isNameExpired(namespaceMap[namespace], nInfo), 'E4');
        mapping(bytes32 => bytes) storage profileValueMap = id2ProfileValueMap[tokenId];
        bytes32[] memory keyList = nInfo.profileKeyList;
        uint256 keyListLen = keyList.length;
        for (uint256 i; i < keyListLen; ) {
            delete profileValueMap[keyList[i]];
            unchecked {
                ++i;
            }
        }
        delete nInfo.profileKeyList;
    }

    //////// ERC721
    function transferFrom(address from, address to, uint256 tokenId) public override nonReentrant {
        require(to != address(0), 'E14');
        (bool bOk, ,) = resolveTokenId(tokenId);
        require(bOk, 'E20');
        super.transferFrom(from, to, tokenId);
    }

    function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory data) public override nonReentrant {
        require(to != address(0), 'E14');
        (bool bOk, ,) = resolveTokenId(tokenId);
        require(bOk, 'E20');
        super.safeTransferFrom(from, to, tokenId, data);
    }

    function tokenURI(uint256 tokenId) public view override returns (string memory) {
        _requireMinted(tokenId);
        return "ipfs://QmUqW8wFpcdYuk5CVFXYvpdsWjz7x9U1Yn6s9NT7f5Re8B";
    }

    // Attention: others can still register the name after burn
    function burn(uint256 tokenId) external nonReentrant {
        require(ownerOf(tokenId) == msg.sender);
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        require(!_isNameExpired(namespaceMap[nInfo.namespace], nInfo), 'E1');
        _removeItem(tokenId);
    }

    function _beforeTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal override {
        if (to != address(0)) {
            (bool bCanReceiveName, uint256 userTokenId) = _canReceiveName(to);
            require(bCanReceiveName, 'E14');
            if (userTokenId > 0) {
                _removeItem(userTokenId);
            }
        }
        super._beforeTokenTransfer(from, to, firstTokenId, batchSize);
    }

    function _afterTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal override {
        if (from != address(0)) {
            addr2TokenIdMap[from] = 0;
        }
        if (to != address(0)) {
            addr2TokenIdMap[to] = firstTokenId;
        }
        super._afterTokenTransfer(from, to, firstTokenId, batchSize);
    }

    //////// Private
    function _canReceiveName(address user) private view returns (bool, uint256) {
        uint256 tokenId = addr2TokenIdMap[user];
        if (tokenId == 0) {
            return (true, 0);
        }
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        return (_isNameExpired(namespaceMap[nInfo.namespace], nInfo), tokenId);
    }

    function _isNameExpired(NamespaceInfo storage nsInfo, NameInfo storage nInfo) private view returns (bool) {
        if (nsInfo.renewalLifetime > 0) {
            return block.timestamp > nInfo.renewalExpiredStamp;
        } else {
            return nsInfo.pingLifetime > 0 ? block.timestamp > nInfo.pingExpiredStamp : false;
        }
    }

    function _assignNameToUser(bytes32 namespace, bytes32 name, address user) private {
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        uint256 tokenId = name2IdMap[namespace][name];
        if (tokenId > 0) {
            NameInfo storage nInfo = id2NameInfoMap[tokenId];
            require(_isNameExpired(nsInfo, nInfo), 'E22');
            if (nsInfo.renewalLifetime > 0) {
                nInfo.renewalExpiredStamp = block.timestamp + nsInfo.renewalLifetime;
            } else {
                nInfo.pingExpiredStamp = block.timestamp + nsInfo.pingLifetime;
            }
            if (ownerOf(tokenId) != user) {
                _transfer(ownerOf(tokenId), user, tokenId);
            }
        } else {
            _mint(user, ++curTokenId);
            bytes32[] memory profileKeyList;
            NameInfo memory nInfo = NameInfo(namespace, name, 0, 0, profileKeyList);
            if (nsInfo.renewalLifetime > 0) {
                nInfo.renewalExpiredStamp = block.timestamp + nsInfo.renewalLifetime;
            } else if (nsInfo.pingLifetime > 0) {
                nInfo.pingExpiredStamp = block.timestamp + nsInfo.pingLifetime;
            }        
            id2NameInfoMap[curTokenId] = nInfo;
            name2IdMap[namespace][name] = curTokenId;
        }
        emit AssignName(namespace, name, user);
    }

    function _removeItem(uint256 tokenId) private {
        NameInfo storage nInfo = id2NameInfoMap[tokenId];
        delete name2IdMap[nInfo.namespace][nInfo.name];
        delete id2NameInfoMap[tokenId];
        _burn(tokenId);
    }

    function _initGenesisNamespace(bytes32 namespace, address beneficiary, uint256 renewalLifetime, uint256 pingLifetime, uint256 renewalPriceParam, uint256 factor1, uint256 factor2) private {
        namespaceList.push(namespace);
        NamespaceInfo storage nsInfo = namespaceMap[namespace];
        if (renewalPriceParam > 0) {
            nsInfo.admin = 0xF0296e8c771b7D422A7E1708324b4260d92D1cEe;
        }
        nsInfo.beneficiary = payable(beneficiary);
        // nsInfo.desc
        nsInfo.startStamp = block.timestamp;
        nsInfo.openStamp = block.timestamp;
        nsInfo.renewalLifetime = renewalLifetime;
        nsInfo.pingLifetime = pingLifetime;
        nsInfo.registerPriceList = [
            factor1 * 2 ether,      factor1 * 2 ether,      factor1 * 1.5 ether,
            factor2 * 0.2 ether,    factor2 * 0.2 ether,    factor2 * 0.15 ether, 
            factor2 * 0.02 ether,   factor2 * 0.02 ether,   factor2 * 0.01 ether, 
            factor2 * 0.005 ether,  factor2 * 0.005 ether,  factor2 * 0.002 ether, 
            factor2 * 0.002 ether,  factor2 * 0.002 ether,  factor2 * 0.001 ether, 
            factor2 * 0.001 ether,  factor2 * 0.001 ether,  factor2 * 0.001 ether, 
            factor2 * 0.001 ether,  factor2 * 0.001 ether,  factor2 * 0.001 ether, 
            factor2 * 0.001 ether,  factor2 * 0.001 ether,  factor2 * 0.001 ether, 
            factor2 * 0.001 ether];
        nsInfo.renewalPriceParam = renewalPriceParam;
        if (renewalPriceParam > 0) {
            nsInfo.bAllowModifyPrice = true;
        }
        /// Custom price names
        bytes32[] storage customNames = nsInfo.customPriceNames;
        mapping(bytes32 => uint256) storage priceMap = customPriceNamesMap[namespace];
        // bitcoin
        customNames.push(0x626974636f696e00000000000000000000000000000000000000000000000000);
        priceMap[0x626974636f696e00000000000000000000000000000000000000000000000000] = renewalPriceParam > 0 ? 2 ether : 50 ether;
        // satoshi
        customNames.push(0x7361746f73686900000000000000000000000000000000000000000000000000);
        priceMap[0x7361746f73686900000000000000000000000000000000000000000000000000] = renewalPriceParam > 0 ? 2 ether : 100 ether;
        // btc
        if (renewalPriceParam > 0) {
            customNames.push(0x6274630000000000000000000000000000000000000000000000000000000000);
            priceMap[0x6274630000000000000000000000000000000000000000000000000000000000] = 2 ether;
        }
    }

    function _initGenesis() private {
        _initGenesisNamespace(0x6200000000000000000000000000000000000000000000000000000000000000, 0x03f4151815694F171B85289Fe8893C24B7876C0f, 0, 2555 days, 0, 10, 10);     // .b namespace. Desc: .b has a great meaning!
        _initGenesisNamespace(0x6400000000000000000000000000000000000000000000000000000000000000, 0x15Fc64076b132aca3daE4870a345C71d9c9eD86C, 0, 0, 0, 10, 10);             // .d namespace. Desc: .d means decentralized.
        _initGenesisNamespace(0x7361740000000000000000000000000000000000000000000000000000000000, 0x96ce96Eaf270Ff9df6dD613c83126135d93Ec041, 365 days, 0, 1001, 1, 5);     // .sat namespace. Desc: With 20 million coins, that gives each coin a value ofabout $10 million. — Hal Finney
        _initGenesisNamespace(0x6269747300000000000000000000000000000000000000000000000000000000, 0xEE6F84C8FFAeAAAd2D9688fF5acE664fb5b21E85, 0, 0, 0, 10, 10);             // .bits namespace. Desc: Why you should register .bits name: <a href="https://x.com/adam3us/status/1662136034514157575?s=20" target="_blank">nostr.com</a>.
        _initGenesisNamespace(0x6e6f737472000000000000000000000000000000000000000000000000000000, 0xF0296e8c771b7D422A7E1708324b4260d92D1cEe, 0, 0, 0, 10, 10);             // .nostr namespace. Desc: nostr is a decentralized social network, visit <a href="https://nostr.com" target="_blank">nostr.com</a> for more info.
    }
}